
*** Theory of totally ordered sets having a smallest
*** next element above any given element.

fth NEXT-TOSET is protecting BOOL .
  sort Elt .
  op _>_ : Elt Elt -> Bool .

  vars X Y Z : Elt .

  eq X > X = false .           *** irreflexivity
  ceq X > Z = true
      if X > Y = true /\ Y > Z = true [nonexec] .
                                 *** transitivity
  ceq X = Y if X > Y = false /\ Y > X = false
                      [nonexec] . *** trychotomy
  
  op eq : Elt Elt -> Bool [comm] .  *** equality 

  eq eq(X,X) = true .
  ceq eq(X,Y) = false if X > Y = true .

  op next : Elt -> Elt .        *** next element
  eq next(X) > X = true .
  ceq (Y > next(X) or eq(Y,next(X))) = true
                               if Y > X = true .
endfth

*** This is a parameterized equational
*** theory describing the
*** functionality needed
*** in the parametric data type of names.

fmod NAME{N :: NEXT-TOSET} is
  sorts  NeSet{N} Set{N} .
  subsort N$Elt < NeSet{N} < Set{N} .



  op mt : -> Set{N} .
  op _+_ : Set{N} Set{N} -> Set{N}
    [ctor assoc comm id: mt] . *** union
  op _+_ : NeSet{N} NeSet{N} -> NeSet{N}
              [ctor assoc comm id: mt] .
  op _\_ : Set{N} Set{N} -> Set{N}  .
                      *** set difference
  op _in_ : NeSet{N} Set{N} -> Bool .
                      *** set membership
                             
  vars X Y : N$Elt .  var S : Set{N} .
  vars T U : NeSet{N} .

  eq X + X = X . 

  eq mt \ S = mt .
  eq T \ mt = T .
  eq (X + S) \ X = S \ X .
  ceq (X + S) \ Y = X + (S \ Y)
                  if eq(X,Y) = false .
  eq (T) \ (Y + U) = (T \ Y) \ U .

  eq X in mt = false .
  eq X in (Y + S) = eq(X,Y) or X in S .
  
  op max : NeSet{N} -> N$Elt .

  eq max(X) = X .
  ceq max(X + Y + S) = max(X + S)
                     if X > Y = true .

  op fresh : NeSet{N} -> N$Elt .
     *** fresh name outside a name set

  eq fresh(S) = next(max(S)) .
endfm



*** The lambda calculus is a rewrite theory
***  whose equational part includes
*** the fv and substitution [_:=_]_ 
*** auxiliary functions, which
*** are sufficiently complete: they go
*** away by equational evaluation. The key
*** computation is rewriting with the beta
*** and eta rules, which takes place MODULO
*** such equations and also modulo the
*** non-exec alpha conversion equation.
***
*** Note that the auxiliary functions fv
*** and  [_:=_]_ are declared with the
*** "frozen" attribute, which forbids
*** rules to be applied below such
*** function symbols.  This
*** is important, since, for example,
***
***(  (\ x . \ y . x) z z' =>beta z but)
***
***(  fv((\ x . \ y . x) z z') = z + z'
  =/= z = fv(z) . )
***
*** That is:
***
***  a. confluence of fv would be lost
***  if beta were treated as an equation; 
***
***  b. even if beta is kept as a rule,
***( fv(U) would become a "moving target")
***  if U could be rewritten by beta,
***  which is prevented by the "frozen"
***   attribute.
***



***  In rewriting logic the specification
*** of frozenness requirements for operators
*** is part of the SEMANTICS of the logic,
*** not just some strategy information
*** Cf. Bruni-Meseguer TCS 2006.
***
*** Note that, since alpha conversion preserves
*** free variables, it can be kept as a nonexec
*** equation without any problems. Note that alpha
*** is the ONLY equation that can be applied
*** between lambda terms.

mod LAMBDA-BETA-ETA{N :: NEXT-TOSET} is
  protecting NAME{N} .
  sort Lambda{N} .
  subsort N$Elt < Lambda{N} . *** subsort of names
  op \_._ : N$Elt Lambda{N} -> Lambda{N} [ctor] .
                            *** lambda abstraction
  op __ : Lambda{N} Lambda{N} -> Lambda{N}
             [ctor gather (E e)] . *** application
  op fv : Lambda{N} -> Set{N} [frozen] .
                                    *** free names
  op [_:=_]_ : N$Elt Lambda{N} Lambda{N} 
          -> Lambda{N} [frozen] . *** substitution

  vars X Y Z : N$Elt .  vars U V W : Lambda{N} .

  eq fv(X) = X .
  eq fv(\ X . U) = fv(U) \ X .
  eq fv(U V) = fv(U) + fv(V) .

  eq  [X := U]X = U .
  ceq [X := U]Y = Y if eq(X,Y) = false .
  eq  [X := U](V W) = ([X := U]V) ([X := U]W) .
  eq  [X := U](\ X . V) = \ X . V .
  ceq [X := U](\ Y . V) = \ Y . [X := U]V



     if eq(X,Y) = false /\ Y in fv(U) = false .
  ceq [X := U](\ Y . V) = \ Z . [X := U][Y := Z]V
      if eq(X,Y) = false /\ Y in fv(U) = true
                  /\ Z := fresh(fv(U) + fv(V))  .

  ceq \ X . U = \ Y . [X := Y]U
     if Y in fv(U) = false [nonexec] . *** Alpha 

  rl [beta] : (\ X . U) V => [X := V]U .
                                        *** Beta 

  crl [eta] :  \ X . (U X) => U
      if X in fv(U) = false .            *** Eta 
endm

*** The following is a possible infinite suppy of
*** names x{0} x{1} x{2} x{3} and so on satisfying the
*** NEXT-TOSET requirements.

fmod VARS is protecting NAT .
  sort Vars .
  op x{_} : Nat -> Vars [ctor] .
  op _>_ : Vars Vars -> Bool .
  op eq : Vars Vars -> Bool .
  op next : Vars -> Vars .
  vars N M : Nat .
  
  eq x{N} > x{M} = N > M .
  eq eq(x{N},x{M}) = (N == M) .
  eq next(x{N}) = x{s(N)} .
endfm

*** Since VARS satisfies the NEX-TOSET theory requirements,
*** we can define the following view:

view Vars from NEXT-TOSET to VARS is



sort Elt to Vars .
endv

*** We can obtained a concrete instance of our
*** name-parametric lambda calculus as follows:

mod LAMBDA-CALCULUS is
  protecting LAMBDA-BETA-ETA{Vars} .
endm

*** Note that the following function definitions are
*** TOTALLY UNNECESSARY in the lambda calculus,
*** since in each such definition we can REPLACE the new
*** name on the left by the corresponding closed
*** lambda term on the right: there is no need for new
*** names to define all these functions;
*** the only reason to use names is for convenience, and
*** to illustrate how various functions can be defined.
*** The representation of the natural numbers is not the
*** usual one by Church numerals, but an alternative
*** representation in Barendregt's Lambda Calculus book.

mod LAMBDA-DEFNS is
  protecting LAMBDA-CALCULUS .

    *** names for function composition (f,g) |-> f ; g
    *** and for squaring function: h |-> h ; h
    
  ops comp square : -> Lambda{Vars} .

  eq comp = \ x{1} . \ x{2} . \ x{0} .(x{2} (x{1} x{0})) .
  eq square = \ x{0} .((comp x{0}) x{0}) .

   *** identity, zero, true, false, if-then-else, Pairing,
   *** 1st and 2nd projections, successor, predecessor,



   *** zero predicate and Turing's fixpoint combinator Yt,
     
  ops I zero tt ff if P p1 p2 suc pred zero? Yt : ->
Lambda{Vars} .
  
  eq I = \ x{0} . x{0} .
  eq zero = \ x{0} . x{0} .
  eq tt = \ x{0} . \ x{1} . x{0} .         
  eq ff = \ x{0} . \ x{1} . x{1} .
  eq if = \ x{0} . \ x{1} . \ x{2} . (x{0} x{1} x{2}) .
  eq P = \ x{1} . \ x{2} . \ x{0} . ((x{0} x{1}) x{2}) .
  eq p1 = \ x{0} .(x{0} tt) .
  eq p2 = \ x{0} .(x{0} ff) .  
  eq suc = \ x{0} . ((P ff) x{0}) .
  eq pred = \ x{0} . (x{0} ff) .
  eq zero? = \ x{0} .(x{0} tt) .
  eq Yt = (\ x{0} . (\ x{1} . (x{1} ((x{0} x{0}) x{1}))))
                     (\ x{0} . (\ x{1} . (x{1} ((x{0} x{0})
x{1})))) .
endm

*** that new names are not needed is illustrated by how
*** they disappear by evaluation.

rew (\ x{0} . \ x{1} . x{0}) x{1} .

rew (\ x{0} . x{1})((\ x{2} . (x{2} x{2})) (\ x{3} . x{3}))
.

rew (\ x{0} .((\  x{1} . (x{1} x{0})) x{2})) x{3} .

rew square .

rew (P zero) tt .

rew p1 .



rew p1 ((P zero) (suc zero)) .

rew p2 ((P zero) (suc zero)) .

rew suc zero .

rew if tt zero (suc zero) .

rew if ff zero (suc zero) .

rew (suc (suc zero)) .

rew (pred (suc zero)) .

rew zero? .

rew zero? (suc zero) .

rew zero? (pred (suc zero)) .

rew ((if (zero? (pred (suc zero)))) zero) ff .

rew [1] Yt .


